Eigenstructure of the Loop Scheme

Denis Zorin, January 1998

In this worksheet we compute the eigenvalues and eigenvectors of the subdivision matrix of the Loop scheme. This analysis repeats the original derivation of Loop from his MS thesis. We also compute eigenvalues and eigenvectors of the modified Loop scheme, which always has largest eigenvalue equal to 1/2.

Utilities

Loop scheme subdivision matrix, eigenvalues, eigenvectors

```
> assume( c, real); additionally( c >= -1 ); additionally( c  <= 1);
> Loop := matrix( [[ (1 - K*alpha)*d,          K*alpha*d,   0,0],
                   [         (3/8)*d,     3/8 + (1/4)*c,   0,0],
                   [        (1/16)*d,     5/8 + (1/8)*c, 1/16, (1/16)*(1+conjugate(omega))],
                   [         (1/8)*d, (3/8)*(1 + omega), 0, 1/8]
                   ]);
```

$$Loop := \begin{bmatrix} (1 - K\,\alpha)\,d & K\,\alpha\,d & 0 & 0 \\ \frac{3}{8}\,d & \frac{3}{8}+\frac{1}{4}\,c\sim & 0 & 0 \\ \frac{1}{16}\,d & \frac{5}{8}+\frac{1}{8}\,c\sim & \frac{1}{16} & \frac{1}{16}+\frac{1}{16}\,\overline{\omega} \\ \frac{1}{8}\,d & \frac{3}{8}+\frac{3}{8}\,\omega & 0 & \frac{1}{8} \end{bmatrix}$$

```
> Loopvar := { omega = c + I*s};
```

$$Loopvar := \{\omega = c\sim + I\,s\}$$

```
> evalues := eigenvals(subs( { d = 0, op(Loopvar)}, evalm(Loop)));
```

$$evalues := 0, \frac{3}{8}+\frac{1}{4}\,c\sim, \frac{1}{16}, \frac{1}{8}$$

```
> EV := max(evalues);
```

$$EV := \frac{3}{8}+\frac{1}{4}\,c\sim$$

```
> LoopZero := map( evalc, subs( { d = 1, c = 1, omega = 1}, evalm(Loop))):
> eigenvals( LoopZero);
```

$$1, \frac{1}{8}, \frac{1}{16}, \frac{5}{8} - K\,\alpha$$

Minimal value of the largest eigenvalue of the first block is 1/4; determine he range for α; α clearly should be less than 5/8K and greater than the following number:

```
> AlphaCrit :=  solve( 5/8 - K*alpha = 3/8 + 1/4*cos(2*Pi/K), alpha );
```

$$AlphaCrit := -\frac{1}{4}\,\frac{-1+\cos\left(2\,\frac{\pi}{K}\right)}{K}$$

```
> evects := subs( s^2 = 1-c^2, [eigenvects(map(evalc,subs( { d = 0, op(Loopvar)}, evalm(Loop))))]);
```

$$evects := \left[\left[\frac{3}{8}+\frac{1}{4}\,c\sim, 1, \left\{\left[0, 1, \frac{1}{2}\,\frac{30\,c\sim+26+4\,c\sim^2}{4\,c\sim^2+9\,c\sim+5}, \frac{3}{2}\,\frac{I\,(-I\,c\sim+s-I)}{c\sim+1}\right]\right\}\right], \left[\frac{1}{8}, 1, \{[0, 0, -I\,(I+I\,c\sim+s), 1]\}\right], \left[\frac{1}{16}, 1, \{[0, 0, 1, 0]\}\right], [0, 1, \{[1, 0, 0, 0]\}]\right]$$

```
> for i from 1 to vectdim(evects) do
    if op(1,op(i, evects)) = EV then v := op(1, op(3, op(i, evects))); fi;
  od;
> LoopEigenvector := map(simplify, map(evalc, eval(v)));
```

$$LoopEigenvector := \left[0, 1, \frac{2\,c\sim+13}{5+4\,c\sim}, \frac{3}{2}\,\frac{c\sim+1+I\,s}{c\sim+1}\right]$$

Drop the first element

```
> LoopEigenvector := vector( [ seq(LoopEigenvector[i], i = 2..4)]);
```

$$LoopEigenvector := \left[1, \frac{2\,c\sim+13}{5+4\,c\sim}, \frac{3}{2}\,\frac{c\sim+1+I\,s}{c\sim+1}\right]$$

Generate code

Modified Loop scheme subdivision matrix, eigenvalues, eigenvectors

This is the expression for all blocks except 0th; the 0th block is the same as for the standard Loop.

```
> ModLoop := matrix( [[ (1 - K*alpha)*d,          K*alpha*d,   0,0],
                     [         (3/8)*d,      (1/2)^m,   0,0],
                     [        (1/16)*d,     5/8 + (1/8)*c, 1/16, (1/16)*(1+conjugate(omega))],
  [         (1/8)*d, (3/8)*(1 + omega), 0, 1/8]
  ]);
```

$$ModLoop := \begin{bmatrix} (1 - K\,\alpha)\,d & K\,\alpha\,d & 0 & 0 \\ \frac{3}{8}\,d & \left(\frac{1}{2}\right)^m & 0 & 0 \\ \frac{1}{16}\,d & \frac{5}{8}+\frac{1}{8}\,c\sim & \frac{1}{16} & \frac{1}{16}+\frac{1}{16}\,\overline{\omega} \\ \frac{1}{8}\,d & \frac{3}{8}+\frac{3}{8}\,\omega & 0 & \frac{1}{8} \end{bmatrix}$$

```
> mevalues := eigenvals(subs( { d = 0, op(Loopvar)}, evalm(ModLoop)));
```

$$mevalues := 0, \left(\frac{1}{2}\right)^m, \frac{1}{16}, \frac{1}{8}$$

The admissable range of α is obvious.

```
> mevects :=  subs( s^2 = 1-c^2, [eigenvects(map(evalc, subs( { d = 0, op(Loopvar) , m = 1}, evalm(ModLoop))))]);
```

$$mevects := \left[[0, 1, \{[1, 0, 0, 0]\}], \left[\frac{1}{16}, 1, \{[0, 0, 1, 0]\}\right], \left[\frac{1}{2}, 1, \{\left[0, 1, \frac{12}{7}+\frac{4}{7}c\sim, I\,(-I\,c\sim+s-I)\right]\}\right], \left[\frac{1}{8}, 1, \{[0, 0, -I\,(I+I\,c\sim+s), 1]\}\right]\right]$$

```
> for i from 1 to vectdim(mevects) do
    if op(1,op(i, mevects)) = 1/2 then mv := op(1, op(3, op(i, mevects))); fi;
  od;
> ModLoopEigenvector := map( evalc, map(simplify, mv));
```

$$ModLoopEigenvector := \left[0, 1, \frac{12}{7}+\frac{4}{7}c\sim, c\sim+1+I\,s\right]$$

Drop the first element

```
> ModLoopEigenvector := vector( [ seq(ModLoopEigenvector[i], i = 2..4)]);
```

$$ModLoopEigenvector := \left[1, \frac{12}{7}+\frac{4}{7}c\sim, c\sim+1+I\,s\right]$$

Generate code